

# Bayesian Dynamic Stochastic General Equilibrium models in Stata 17

David Schenck

Senior Econometrician  
Stata

2021 Belgian Stata Conference  
June 7, 2021

# Bayesian econometrics in Stata

- Stata 17 introduces Bayesian estimation of a variety of time–series and panel–data econometric models
  - `bayes: dsge`
  - `bayes: dsgenl`
  - `bayes: var`
  - `bayesirf`
  - `bayesfcst`
  - `bayes: xt`

# Outline

- DSGE models, dsge, and bayes: dsge
- AR(1) model
- Linear New Keynesian model
- Nonlinear stochastic growth model

# DSGE models

- A DSGE model is a system of equations that describes an economy.
- A model consists of three kinds of variables:
  - Control variables, whose values are determined by the system of equations each period
  - State variables, which are fixed at the beginning of any given period but whose laws of motion are part of the system of equations
  - Stochastic shocks, which drive the system
- DSGE models come from economic theory. Theories are forward-looking, so equations are forward-looking.
- Models are used for policy analysis: explore different policy alternatives or how different parameter values affect model outcomes

## An example model

- Suppose we wished to model the effect of monetary policy on macroeconomic variables
- We model relationships among output, inflation, and the interest rate
- These variables are linked to state variables representing monetary shocks, and perhaps other factors
- State variables, in turn, are driven by shocks

## An example model

- The following system of equations is a DSGE model:

$$x_t = E_t x_{t+1} - (r_t - E_t \pi_{t+1} - z_t) \quad (\text{IS})$$

$$\pi_t = \beta E_t \pi_{t+1} + \kappa x_t \quad (\text{Phillips Curve})$$

$$r_t = \frac{1}{\beta} \pi_t + w_t \quad (\text{Taylor Rule})$$

$$w_{t+1} = \rho_w w_t + e_{t+1}$$

$$z_{t+1} = \rho_z z_t + \varepsilon_{t+1}$$

- Control variables:  $(x_t, \pi_t, r_t)$
- State variables:  $(w_t, z_t)$
- Stochastic shock:  $(e_t, \varepsilon_t)$
- Parameters:  $(\kappa, \beta, \rho_w, \rho_z, \sigma_e^2, \sigma_\varepsilon^2)'$

# DSGE models in Stata

- Model:

$$x_t = E_t x_{t+1} - (r_t - E_t \pi_{t+1} - z_t) \quad (\text{IS})$$

$$\pi_t = \beta E_t \pi_{t+1} + \kappa x_t \quad (\text{Phillips Curve})$$

$$r_t = \frac{1}{\beta} \pi_t + w_t \quad (\text{Taylor Rule})$$

$$w_{t+1} = \rho_w w_t + e_{t+1}$$

$$z_{t+1} = \rho_z z_t + \varepsilon_{t+1}$$

- In Stata: `dsge` (*model\_equations*)

```
. dsge (x = F.x - (r - F.p - z) , unobserved ) ///
      (p = {beta}*F.p + {kappa}*x ) ///
      (r = (1/{beta})*p + w ) ///
      (F.z = {rhoz}*z , state ) ///
      (F.w = {rhow}*w , state )
```


## Two views on the impact of monetary shocks

```
. quietly webuse usmacro2
. matrix param1 =          (0.1, 0.5, 0.9, 0.7)
. matrix colnames param1 = kappa beta rhoz rhow
. matrix param2 =          (1, 0.5, 0.9, 0.7)
. matrix colnames param2 = kappa beta rhoz rhow
.
. quietly irf set dsge_irf.irf, replace
.
. quietly dsge (x = F.x - (r - F.p - z) , unobserved )    ///
> (p = {beta}*F.p + {kappa}*x )    ///
> (r = (1/{beta})*p + w )    ///
> (F.z = {rhoz}*z , state )    ///
> (F.w = {rhow}*w , state ) ,    ///
> from(param1) solve
. quietly irf create keynes
.
. quietly dsge (x = F.x - (r - F.p - z) , unobserved )    ///
> (p = {beta}*F.p + {kappa}*x )    ///
> (r = (1/{beta})*p + w )    ///
> (F.z = {rhoz}*z , state )    ///
> (F.w = {rhow}*w , state ) ,    ///
> from(param2) solve
. quietly irf create classical
```



# Two views on the impact of monetary shocks

## Impulse responses



# Two views on the impact of monetary shocks

## Maximum likelihood estimation

```
. dsge (x = F.x - (r - F.p - z) , unobserved ) ///
> (p = {beta}*F.p + {kappa}*x ) ///
> (r = (1/{beta})*p + w ) ///
> (F.z = {rhoz}*z , state ) ///
> (F.w = {rhow}*w , state ) , nolog
```

DSGE model

Sample: 1955q1 thru 2015q4  
Log likelihood = -753.57131

Number of obs = 244

	Coefficient	Std. err.	z	P> z	[95% conf. interval]	
/structural						
beta	.5146664	.0783491	6.57	0.000	.3611051	.6682278
kappa	.1659059	.0474074	3.50	0.000	.0729891	.2588226
rhoz	.9545256	.0186424	51.20	0.000	.9179872	.991064
rhow	.7005484	.0452604	15.48	0.000	.6118396	.7892572
sd(e.z)	.6211222	.1015082			.4221699	.8200745
sd(e.w)	2.318207	.3047457			1.720916	2.915497

# Bayesian DSGE models

- Estimation of DSGE models has shifted towards the Bayesian approach in the past 15 years
- Substantive:
  - Priors reflect genuine prior beliefs about the distribution of parameter values
  - Priors allow the incorporation of other evidence that is hard to incorporate into the likelihood, e.g. micro-evidence on price adjustment
- Technical:
  - Mapping from structural parameters to solution matrices is highly nonlinear, and identification issues are pervasive
  - Priors can aid in isolating identification problems
  - Priors useful for imposing bounds on parameters

# Bayesian analysis and the bayes prefix

- The bayes: prefix allows for Bayesian estimation of likelihood-based models
- Just attach bayes: to the existing Stata command
  - **bayes:** regress y x
  - default priors provided (can be overwritten)
- For DSGE models:
  - bayes, **prior(prior\_spec)**: dsge (*model\_equations*)
  - some priors are required

# Outline

- DSGE models, dsge, and bayes: dsge
- AR(1) model
- Linear New Keynesian model
- Nonlinear stochastic growth model

# US macro data


```
. webuse usmacro2
(Federal Reserve Economic Data - St. Louis Fed, 2017-01-15)
. describe
Contains data from https://www.stata-press.com/data/r17/usmacro2.dta
Observations:      244      Federal Reserve Economic Data -
                        St. Louis Fed, 2017-01-15
Variables:         11      1 May 2020 17:52
                        (_dta has notes)
```

Variable name	Storage type	Display format	Value label	Variable label
daten	int	%td		Numeric (daily) date
year	int	%9.0g		Year
quarter	byte	%9.0g		Quarter
dateq	int	%tq		Date (quarters)
y	double	%10.0g		Growth rate of real GDP (GDPC96)
p	double	%10.0g		Growth rate of prices (GDPDEF)
r	double	%10.0g		Federal funds rate (FEDFUNDS)
c	double	%10.0g		Growth rate of consumption (PCECC96)
n	double	%10.0g		Growth rate of hours worked (HOANBS)
i	double	%10.0g		Corporate bond interest rate (AAA)
e	double	%10.0g		Percentage change in US exchange rate (TWEXBMTH)

Sorted by: dateq

# GDP growth rate

```
. tsline y
```



# An AR(1) model

## Model

- Model:

$$y_t = \rho y_{t-1} + u_t$$

- State-space formulation:

$$y_t = z_t \quad \text{(Observation equation)}$$

$$z_t = \rho z_{t-1} + u_t \quad \text{(State transition equation)}$$

- Stata specification:

```
. dsge (y=z) (f.z = {rho}*z, state)
```

- With the bayes: prefix:

```
. bayes, prior({rho}, uniform(-1,1)): dsge (y=z) (f.z = {rho}*z, state)
```



# An AR(1) model

## Output

```
. bayes, prior({rho}, uniform(-1,1)) rseed(20) : ///  
>      dsge (y = z) (f.z = {rho}*z, state)  
note: initial parameter vector set to means of priors.
```

Burn-in ...

Simulation ...

Model summary

---

Likelihood:

y ~ dsgell({rho},{sd(e.z)})

Priors:

{rho} ~ uniform(-1,1)

{sd(e.z)} ~ igamma(.01,.01)

---

*(Output continues on next slide)*

# An AR(1) model

## Output II

Bayesian linear DSGE model  
Random-walk Metropolis-Hastings sampling

Sample: 1955q1 thru 2015q4

Log marginal-likelihood = -648.62049


MCMC iterations = 12,500  
Burn-in = 2,500  
MCMC sample size = 10,000  
Number of obs = 244  
Acceptance rate = .266  
Efficiency: min = .1109  
                  avg = .1181  
                  max = .1253

	Mean	Std. dev.	MCSE	Median	Equal-tailed [95% cred. interval]	
rho	.3361959	.0592612	.001674	.3351618	.2176139	.4571248
sd(e.z)	3.340431	.150282	.004513	3.336891	3.060339	3.649831

# An AR(1) model

## Diagnostics

```
. bayesgraph diagnostics {rho}
```



# An AR(1) model

## More diagnostics

```
. bayesstats ess
```

```
Efficiency summaries      MCMC sample size =    10,000
                          Efficiency:  min =      .1109
                              avg =      .1181
                              max =      .1253
```

	ESS	Corr. time	Efficiency
rho	1252.97	7.98	0.1253
sd(e.z)	1108.99	9.02	0.1109

# Summary

- Basic syntax of DSGE models:

```
. dsge (model_equations)
```

- Basic syntax of Bayesian DSGE models:

```
. bayes, prior(prior_spec) : dsge (model_equations)
```

- Parameter estimation
- Postestimation diagnostics

# Outline

- DSGE models, dsge, and bayes: dsge
- AR(1) model
- **Linear New Keynesian model**
- Nonlinear stochastic growth model

## Linear DSGE models

- We now return to a small, fully-featured DSGE model
- Equations:

$$x_t = E_t x_{t+1} - (r_t - E_t \pi_{t+1} - z_t) \quad (\text{IS})$$

$$\pi_t = \beta E_t \pi_{t+1} + \kappa x_t \quad (\text{Phillips Curve})$$

$$r_t = \frac{1}{\beta} \pi_t + w_t \quad (\text{Taylor Rule})$$

$$w_{t+1} = \rho_w w_t + e_{t+1}$$

$$z_{t+1} = \rho_z z_t + \varepsilon_{t+1}$$

- Control equations for output gap, inflation, interest rate
- State equations for monetary and IS disturbances (AR(1))
- Equations are linear in variables, nonlinear in parameters
- Forward-looking elements in the control equations
- Shocks flow into state variables, then into control variables

# Linearized New Keynesian model

## Data

```
. webuse usmacro2
(Federal Reserve Economic Data - St. Louis Fed, 2017-01-15)

. describe
Contains data from https://www.stata-press.com/data/r17/usmacro2.dta
Observations:      244      Federal Reserve Economic Data -
                        St. Louis Fed, 2017-01-15
Variables:         11      1 May 2020 17:52
                        (_dta has notes)
```


Variable name	Storage type	Display format	Value label	Variable label
daten	int	%td		Numeric (daily) date
year	int	%9.0g		Year
quarter	byte	%9.0g		Quarter
dateq	int	%tq		Date (quarters)
y	double	%10.0g		Growth rate of real GDP (GDPC96)
p	double	%10.0g		Growth rate of prices (GDPDEF)
r	double	%10.0g		Federal funds rate (FEDFUNDS)
c	double	%10.0g		Growth rate of consumption (PCECC96)
n	double	%10.0g		Growth rate of hours worked (HOANBS)
i	double	%10.0g		Corporate bond interest rate (AAA)
e	double	%10.0g		Percentage change in US exchange rate (TWEXBMTH)



# Linearized New Keynesian model

## Data

```
. tsline p r, legend(rows(2))
```



# Linear New Keynesian model

## Model specification

```
. bayes, prior({beta}, beta(10, 10)) prior({kappa}, beta(30, 70)) ///
> prior({rhow}, beta(10, 10)) prior({rhoz}, beta(35,15)) ///
> rseed(17) dots: ///
> dsge (x = F.x - (r - F.p - z) , unobserved ) ///
> (p = {beta}*F.p + {kappa}*x ) ///
> (r = (1/{beta})*p + w ) ///
> (F.z = {rhoz}*z , state ) ///
> (F.w = {rhow}*w , state )
note: initial parameter vector set to means of priors.

Burn-in 2500 aaaaaaaaaa1000.....2000..... done
Simulation 10000 .....1000.....2000.....3000.....4000.....5
> 000.....6000.....7000.....8000.....9000.....10000 done
```

# Linear New Keynesian model

## Estimation header

### Model summary

---

#### Likelihood:

```
p r ~ dsgell({beta},{kappa},{rhoz},{rhow},{sd(e.z)},{sd(e.w)})
```

#### Priors:

```
    {beta} ~ beta(10,10)
    {kappa} ~ beta(30,70)
    {rhoz} ~ beta(35,15)
    {rhow} ~ beta(10,10)
    {sd(e.z) sd(e.w)} ~ igamma(.01,.01)
```

---

# Linear New Keynesian model

## Estimation output

---

```
Bayesian linear DSGE model
Random-walk Metropolis-Hastings sampling

Sample: 1955q1 thru 2015q4

Log marginal-likelihood = -784.87902

MCMC iterations = 12,500
Burn-in = 2,500
MCMC sample size = 10,000
Number of obs = 244
Acceptance rate = .1523
Efficiency: min = .01375
              avg = .01964
              max = .02488
```

---


	Mean	Std. dev.	MCSE	Median	Equal-tailed [95% cred. interval]	
beta	.4731495	.0506063	.003631	.4739432	.3692667	.5735173
kappa	.261336	.0374749	.002812	.2595999	.1971056	.341189
rhoz	.9084647	.0151727	.000962	.9083686	.8783911	.9359358
rhow	.6309735	.033607	.002866	.6322336	.563363	.6972567
sd(e.z)	.7389675	.0738833	.005184	.7357219	.60302	.8981278
sd(e.w)	2.53286	.2537029	.017219	2.495344	2.128193	3.108239

---

# Linear New Keynesian model

## Diagnostics


```
. bayesgraph diagnostics kappa
```



# Linear New Keynesian model

## Prior-posterior plot

```
. bayesgraph kdensity {kappa}, lcolor(red)      ///  
> addplot(function Prior = betaden(30,70, x),  ///  
> legend(on label(1 "Posterior")) lcolor(blue))
```



# Linear New Keynesian model

## Diagnostics

```
. bayesstats ess
```

```
Efficiency summaries      MCMC sample size =    10,000
                          Efficiency:  min =    .01375
                                      avg =    .01964
                                      max =    .02488
```

	ESS	Corr. time	Efficiency
beta	194.26	51.48	0.0194
kappa	177.56	56.32	0.0178
rhoz	248.80	40.19	0.0249
rhow	137.55	72.70	0.0138
sd(e.z)	203.12	49.23	0.0203
sd(e.w)	217.08	46.07	0.0217

# Linear New Keynesian model

## Impulse responses

- One key object of interest in DSGE analysis is the impulse response function (IRF)
- IRFs plot the response of model variables to unexpected shocks to a state variable
- A type of counterfactual analysis
- Answers to policy questions: what is the effect of a change in monetary policy (or fiscal policy, oil prices, etc) on variables of interest?



# Linear New Keynesian model

## Impulse responses

- Stata has the `irf` suite of commands to create, plot, and make tables of IRFs
- Available after `dsge`, `var`, and some other time-series commands
- Now, this suite is extended to work after Bayesian estimation with `bayesirf`
  - `bayesirf set` to set a results file
  - `bayesirf create` to compute IRFs
  - `bayesirf graph` for plots
  - `bayesirf table` for tables
  - and other utility commands


# Linear New Keynesian model

## Impulse responses

```
. bayesirf set nkirf.irf, replace  
(file nkirf.irf created)  
(file nkirf.irf now active)  
. bayesirf create model1  
(file nkirf.irf updated)  
. bayesirf graph irf, impulse(w) response(x p r w) byopts(yrescale)
```

# Linear New Keynesian model

## Impulse responses



# Linear New Keynesian model

## Underidentified model

- Bayesian estimation can be performed even when the model parameters are not classically identified
- Identification then comes off the prior
- Consider the model:

$$x_t = E_t x_{t+1} - \sigma(r_t - E_t \pi_{t+1} - z_t) \quad (\text{IS})$$

$$\pi_t = \beta E_t \pi_{t+1} + \kappa x_t \quad (\text{Phillips Curve})$$

$$r_t = \frac{1}{\psi} \pi_t + w_t \quad (\text{Taylor Rule})$$

$$w_{t+1} = \rho_w w_t + e_{t+1}$$

$$z_{t+1} = \rho_z z_t + \varepsilon_{t+1}$$

- New parameters ( $\sigma, \psi$ ) not identified just from data on inflation and interest rate
- We can bring in prior information to aid estimation

# Linear New Keynesian model

## Underidentified model: setup

```
. bayes, prior({beta}, beta(95, 5)) prior({kappa}, beta(30,70)) ///
> prior({sigma}, beta(10,90)) prior({psi}, beta(67,33)) ///
> prior({rhoz}, beta(10, 10)) prior({rhoz}, beta(35,15)) ///
> rseed(17) dots burnin(5000) mcmcsize(30000): ///
> dsge (x = F.x - {sigma}*(r - F.p - z) , unobserved ) ///
> (p = {beta}*F.p + {kappa}*x ) ///
> (r = (1/{psi})*p + w ) ///
> (F.z = {rhoz}*z , state ) ///
> (F.w = {rhoz}*w , state )
note: initial parameter vector set to means of priors.
```

```
Burn-in 5000 aaaaaaaaa1000aaaaaaaaa2000aaaa.....3000.....4000.....5000
> done
Simulation 30000 .....1000.....2000.....3000.....4000.....5
> 000.....6000.....7000.....8000.....9000.....10000.....
> .11000.....12000.....13000.....14000.....15000.....16000.
> .....17000.....18000.....19000.....20000.....21000.....
> .22000.....23000.....24000.....25000.....26000.....27000.
> .....28000.....29000.....30000 done
```

# Linear New Keynesian model

Underidentified model: header output

## Model summary

---

### Likelihood:

```
p r ~ dsgell({sigma},{beta},{kappa},{psi},{rhoz},{rhow},{sd(e.z)},{sd(e.w)})
```

### Priors:

```
{sigma} ~ beta(10,90)
```

```
{beta} ~ beta(95,5)
```

```
{kappa} ~ beta(30,70)
```

```
{psi} ~ beta(67,33)
```

```
{rhoz} ~ beta(35,15)
```

```
{rhow} ~ beta(10,10)
```

```
{sd(e.z) sd(e.w)} ~ igamma(.01,.01)
```

---

# Linear New Keynesian model

Underidentified model: estimation output

Bayesian linear DSGE model  
Random-walk Metropolis-Hastings sampling

Sample: 1955q1 thru 2015q4

Log marginal-likelihood = -787.73905

MCMC iterations = 35,000  
Burn-in = 5,000  
MCMC sample size = 30,000  
Number of obs = 244  
Acceptance rate = .1741  
Efficiency: min = .005331  
                  avg = .01032  
                  max = .01974

	Mean	Std. dev.	MCSE	Median	Equal-tailed [95% cred. interval]	
sigma	.1443227	.0292876	.001318	.1432416	.088498	.2043906
beta	.9547238	.0203592	.001276	.9576523	.9077723	.9848212
kappa	.3419745	.0457376	.003318	.3415295	.2517864	.4357346
psi	.6527897	.043041	.003403	.6529768	.5686343	.7351054
rhoz	.9078086	.0157278	.00091	.9080455	.8749843	.9369648
rhow	.7546737	.0269813	.001109	.7541327	.7017534	.8085894
sd(e.z)	.6048148	.0950875	.005623	.5951787	.4475909	.8237128
sd(e.w)	1.955303	.1265823	.008904	1.948905	1.734296	2.230285

# Linear New Keynesian model

Underidentified model: summaries of functions of parameters

```
. bayesstats summary (1/{psi})
```

```
Posterior summary statistics
```

```
MCMC sample size = 30,000
```

```
expr1 : 1/{psi}
```


	Mean	Std. dev.	MCSE	Median	Equal-tailed [95% cred. interval]	
expr1	1.538681	.1035836	.008064	1.531448	1.360349	1.7586



# Linear New Keynesian model

Underidentified model: diagnostics of functions of parameters

```
. bayesgraph diagnostics (1/{psi})
```







# Linear New Keynesian model

## Underidentified model: prior/posterior graph setup

```
. bayesgraph kdensity {kappa}, lcolor(red) ///
>     addplot(function Prior=betaden(30,70,x), ///
>     legend(on label(1 "Posterior")) lcolor(blue)) name(kappa) nodraw
.
. bayesgraph kdensity {sigma}, lcolor(red) ///
>     addplot(function Prior=betaden(10,90,x), ///
>     legend(on label(1 "Posterior")) lcolor(blue)) name(sigma) nodraw
.
. bayesgraph kdensity {beta}, lcolor(red) ///
>     addplot(function Prior=betaden(95,5,x), ///
>     legend(on label(1 "Posterior")) lcolor(blue)) name(beta) nodraw
.
. bayesgraph kdensity {psi}, lcolor(red) ///
>     addplot(function Prior=betaden(67,33,x), ///
>     legend(on label(1 "Posterior")) lcolor(blue)) name(psi) nodraw
.
. graph combine kappa sigma beta psi
```

# Linear New Keynesian model

Underidentified model: prior/posterior graphs



# Outline

- DSGE models, dsge, and bayes: dsge
- AR(1) model
- Linear New Keynesian model
- **Nonlinear stochastic growth model**

## Nonlinear DSGE models

- Linear DSGE models are linear in variables, potentially nonlinear in parameters
- Nonlinear DSGE models are nonlinear in both variables and parameters
- `dsgen1` estimates the parameters of nonlinear DSGE models by first-order approximation (linearization)
- `bayes`: `dsgen1` does the same, with Bayesian methods

# The stochastic growth model

## Equations

$$\frac{1}{c_t} = \beta E_t \left[ \left( \frac{1}{c_{t+1}} \right) (1 + r_{t+1} - \delta) \right] \quad \text{(Consumption)}$$

$$r_t = \alpha \frac{y_t}{k_t} \quad \text{(Capital demand)}$$

$$y_t = z_t k_t^\alpha \quad \text{(Production function)}$$

$$k_{t+1} = y_t - c_t + (1 - \delta)k_t \quad \text{(Capital accumulation)}$$

$$\ln z_{t+1} = \rho \ln z_t + e_{t+1}$$

# The stochastic growth model

## Setup

```
. bayes, prior({alpha}, beta(30,70)) prior({beta}, beta(95,5)) ///
> prior({delta}, beta(25,975)) prior({rho}, beta(5, 3)) ///
> rseed(17) burnin(5000) dots : ///
> dsngenl (1/c      = {beta}*(1/f.c)*(1+f.r-{delta})) ///
>          (r      = {alpha}*y/k) ///
>          (y      = z*k^{alpha}) ///
>          (f.k    = y - c + (1-{delta})*k) ///
>          (ln(f.z) = {rho}*ln(z)) ///
>          , exostate(z) endostate(k) observed(y) unobserved(c r)
note: initial parameter vector set to means of priors.

Burn-in 5000 aaaaaaaaa1000aaaaaaaaa2000aaaaaaaaa3000aaaaaaa...4000.....5000
> done
Simulation 10000 .....1000.....2000.....3000.....4000.....5
> 000.....6000.....7000.....8000.....9000.....10000 done
```

# The stochastic growth model

## Header output

### Model summary

---

#### Likelihood:

```
y ~ dsgell({beta},{delta},{alpha},{rho},{sd(e.z)})
```

#### Priors:

```
{beta} ~ beta(95,5)  
{delta} ~ beta(25,975)  
{alpha} ~ beta(30,70)  
{rho} ~ beta(5,3)  
{sd(e.z)} ~ igamma(.01,.01)
```

---



# The stochastic growth model

## Estimation output

```
Bayesian first-order DSGE model
Random-walk Metropolis-Hastings sampling

Sample: 1955q1 thru 2015q4

Log marginal-likelihood = -649.82949

MCMC iterations = 15,000
Burn-in = 5,000
MCMC sample size = 10,000
Number of obs = 244
Acceptance rate = .2506
Efficiency: min = .04563
              avg = .04999
              max = .05552
```

	Mean	Std. dev.	MCSE	Median	Equal-tailed [95% cred. interval]	
beta	.9554009	.0194803	.00086	.9582636	.9104263	.9847218
delta	.0250477	.0048846	.000207	.0247132	.0163193	.0357851
alpha	.2962864	.0442748	.002073	.2958201	.2122214	.3837115
rho	.3064437	.0584439	.002654	.3047101	.1939408	.422685
sd(e.z)	3.359195	.152429	.006891	3.360512	3.068458	3.658345





# The stochastic growth model

## Prior/posterior graphs

```
. bayesgraph kdensity {alpha}, lcolor(red) ///
>     addplot(function Prior=betaden(30,70,x), ///
>     legend(on label(1 "Posterior")) lcolor(blue)) name(alpha) nodraw
.
. bayesgraph kdensity {beta}, lcolor(red) ///
>     addplot(function Prior=betaden(95, 5,x), ///
>     legend(on label(1 "Posterior")) lcolor(blue)) name(beta) nodraw
.
. bayesgraph kdensity {delta}, lcolor(red) ///
>     addplot(function Prior=betaden(25,975,x), ///
>     legend(on label(1 "Posterior")) lcolor(blue)) name(delta) nodraw
.
. bayesgraph kdensity {rho}, lcolor(red) ///
>     addplot(function Prior=betaden(5,3,x), ///
>     legend(on label(1 "Posterior")) lcolor(blue)) name(rho) nodraw
.
. graph combine alpha beta delta rho
```

# The stochastic growth model

## Prior/posterior graphs




# The stochastic growth model

## Impulse responses

```
. bayesirf set stochmodel, replace  
(file stochmodel.irf created)  
(file stochmodel.irf now active)  
. bayesirf create model1  
(file stochmodel.irf updated)  
. bayesirf graph irf, impulse(z) byopts(yrescale)
```

# The stochastic growth model

## Impulse responses



# Final thoughts

- the `bayes:` prefix now supports `dsge` and `dsgen1`
- Bayesian impulse response functions supported with `bayesirf`

Thank you!